Advanced GORM

1

Advanced GORM

- Dynamic finders
- Hibernate Query Language (HQL)
- Criteria Builder
- Where Queries
- Named Queries
- Events / Automatic timestamping

2

Dynamic Finders

3

Dynamic Finders

- Synthesized at runtime (not compiled)
- Made of up to two properties, boolean operators and comparators
- Additional parameters can be passed for pagination and sorting

4

Dynamic Finders

- findAllBy
  - returns list of objects
- findBy
  - returns single object

5

Dynamic Finders

```
findByName('mike')
findAllByName('mike')
findAllByNameAndAccountNumber('mike', '123')
findAllByNameOrAccountNumber('mike', '789')
findAllByNameLike('J%')

findAllByNameIlike('j%',
    [sort:'name'])

findAllByNameIlike('j%',
    [order:'desc',sort:'name'])

findAllByNameIlike('j%',
    [order:'desc', sort:'name',
    max:10, offset:0])
```

6

---

meta programming

7

---

modify runtime behavior

8

```
class Person {
    String name
    Integer age
}
Person.findByName('Mike')
```

9

---

metaClass

10

```groovy
def fakeDb = [
    new Person(name:'Mike', gender:'M'),
    new Person(name:'Robin', gender:'F')]

class Person {
    String name
    String gender
    String toString() {name}
}

// define a metaClass method on Person
Person.metaClass.static.findByName =
{ String name->
    return fakeDb.findAll {it.name == name}
}

Person.findByName('Robin')
```

11

---

invokeMethod

12

```groovy
class AddressBook {
  def people = [
    new Person(name:'Mike', gender:'M'),
    new Person(name:'Robin', gender:'F')]

  def invokeMethod(String name, args) {
    def propToFind = name - 'findBy'
    people.findAll {
      it[propToFind.toLowerCase()] == args[0]
    }
  }
}

def ab = new AddressBook()
ab.findByName('Robin')
ab.findByGender('F')
```

13

---

methodMissing

14

```
1  class AddressBook {
2    static people = [
3        new Person(name: 'Mike', gender: 'M'),
4        new Person(name: 'Robin', gender: 'F')]
5
6    // intercept
7    def methodMissing(String name, args) {
8      println 'inside methodMissing'
9      def impl = { Object[] theArgs ->
10       def propToFind = name - 'findBy'
11       people.findAll {
12         it[propToFind.toLowerCase()] == args[0]
13       }
14     }
15     // cache
16     AddressBook.metaClass."${name}" = impl
17     // invoke
18     return impl(args)
19   }
20 }
21 def ab = new AddressBook()
22 ab.findByName('Robin')
```

intercept

15

```
1  class AddressBook {
2    static people = [
3        new Person(name: 'Mike', gender: 'M'),
4        new Person(name: 'Robin', gender: 'F')]
5
6    // intercept
7    def methodMissing(String name, args) {
8      println 'inside methodMissing'
9      def impl = { Object[] theArgs ->
10       def propToFind = name - 'findBy'
11       people.findAll {
12         it[propToFind.toLowerCase()] == args[0]
13       }
14     }
15     // cache
16     AddressBook.metaClass."${name}" = impl
17     // invoke
18     return impl(args)
19   }
20 }
21 def ab = new AddressBook()
22 ab.findByName('Robin')
```

cache

16

```groovy
class AddressBook {
  static people = [
      new Person(name: 'Mike', gender: 'M'),
      new Person(name: 'Robin', gender: 'F')]

  // intercept
  def methodMissing(String name, args) {
    println 'inside methodMissing'
    def impl = { Object[] theArgs ->
      def propToFind = name - 'findBy'
      people.findAll {
        it[propToFind.toLowerCase()] == args[0]
      }
    }
    // cache
    AddressBook.metaClass."${name}" = impl
    // invoke
    return impl(args)
  }
}
def ab = new AddressBook()
ab.findByName('Robin')
```

invoke

17

---

summary

18

- First Time
  - ab.findByName('Robin')
  - invokeMethod
  - methodMissing
  - method cached
  - method invokved
- Second Time
  - ab.findByName('Robin')
  - invokeMethod
  - method invoked

19

Problem

- Something more advanced than a dynamic finder

- Dynamic finders are great but can lead to programmer laziness and performance issues

20

Example

Find all customers with gold service level and more than 5 incidents

```
def gold = ServiceLevel.findByName('Gold')

List goldCustomers =
  Customer.findAllByServiceLevel(gold)

List moreThanFive =
  goldCustomers.findAll { cust->
    cust.incidents.size() > 5
}
```

21

```
def gold = ServiceLevel.findByName('Gold')
// select * from service_level where name = ?

List goldCustomers =
  Customer.findAllByServiceLevel(gold)
// select * from customer where service_level_id = ?

List moreThanFive =
  goldCustomers.findAll { cust->
    // iterates through each customer in the list

    cust.incidents.size() > 5
    // select * from incident where customer_id = ?
    // (for each customer!)
}
```

22

HQL

23

HQL

- Hibernate Query Language
- SQL-like query language using domain and property names rather than DB names

```
Customer.findAll(
  "from Customer as c where c.name = 'Mike'")

Customer.findAll(
  "from Customer as c where c.name = ?", ['Mike'])

Customer.findAll(
  "from Customer as c where c.name = :name",
    [name:'mike'])
```

24

---

methods

- find
- findAll
- executeQuery
- executeUpdate

25

---

find / findAll

- find: returns a single object (first found)
- findAll: returns a list of objects

26

```
Customer.findAll(
  "from Customer as c
   inner join fetch c.address
   where c.serviceLevel.name = ?
   and size(c.incidents) > 5",
    ['Gold'])



Customer.findAll(
  "from Customer as c            <--- class name
   inner join fetch c.address    <--- property
   where c.serviceLevel.name = ? <--- parameter
   and size(c.incidents) > 5",
    ['Gold'])                    <--- parameter value
```

27

---

executeQuery

- doesn't return a domain class
- good for getting a subset of data
  - loading a single column
  - count, min, max, etc

28

```
Address.executeQuery(
    "select distinct state, count(*)
    from Address
    group by state")
```

Returns a list of results, where each result is a list itself, e.g.:

```
[ [MN, 5], [WI, 10] ]
```

29

---

executeUpdate

- Data Manipulation
  - UPDATE
  - DELETE
- e.g. update all silver statuses to platinum
- delete all accounts with no activity in the 90 days

30

```
Customer.executeUpdate(
    '''update Customer c
    set c.serviceLevel = :newSl
    where c.serviceLevel = :oldSl''',
    [
     newSl:ServiceLevel.findByName('SuperAwesome'),
     oldSl: ServiceLevel.findByName('Gold')
    ]
)
```

31

Criteria

32

Criteria

- Grails provides a Hibernate Criteria Builder

- Useful for forming dynamic queries
- Two methods:
  - createCriteria
  - withCriteria - inline criteria builder

33

---

restrictions

- `eq` - equal
- `ilike` - case insensitive like (wildcard: %) like - case sensitive like (wildcard: %)
- `in` - in a list
- `isNull`
- `lt`, `le` - less than; less than or equal
- `gt`, `ge` - greater than; greater than or equal

34

---

query methods

- list - (default) returns all matching rows
- listDistinct - return a distinct set of results
- get - retrieve one row (useful for projections)
- scroll - returns a scrollable result set

35

Criteria

```
Customer.withCriteria {
    eq('name', 'mike')
}

// is the same as

def c = Customer.createCriteria()
c {
    eq('name', 'mike')
}
```

36

Criteria

```groovy
def c = Customer.createCriteria() c{
    or {
        eq('name', 'mike')
        eq('accountNumber', '789')
    }
}

// is the same as

findAllByNameOrAccountNumber('mike', '789')
```

37

Criteria

```groovy
def customerByName(nameToFind) {
    def c = Customer.createCriteria()
    return c.list() {
        eq('name', nameToFind)
    }
}

// is the same as

def customerByName(nameToFind) {
    return Customer.withCriteria {
        eq('name', nameToFind)
    }
}
```

38

Example

- simple: find all customers by name
- find all customers with gold service level and more than 5 incidents
- Advanced Search function
    - name
    - account number
    - state

39

```groovy
def customersByName(String theName) {
    def c = Customer.createCriteria()
    return c.list() {
        eq('name', theName)
    }
}
```

simple: find all customers by name

40

```
def goldWithFiveCriteria() {
    def c = Customer.createCriteria()
    return c.list() {
        serviceLevel {
            eq('name', 'Gold')
        }
        sizeGt('incidents', 5)
    }
}
```

find all customers with gold service level and more than 5 incidents

41

```
def search(accountNumber, name, state) {
  def c = Customer.createCriteria()
  return c.list {
    or {
      if (accountNumber) {
        ilike('accountNumber', "${accountNumber}%")
      }
      if (name) {
        ilike('name', "${name}%")
      }
    }
    if (state) {
      address {
        eq('state', state)
      }
    }
  }
}
```

"advanced" search function

42

Where Queries

43

Where Queries

- Define a query using boolean logic

Where Query

```
Customer.where {
    serviceLevel.name == 'Gold'
    &&
    incidents.size() > 5
}.list()
```

find all customers with gold service level and more than 5 incidents

Combining Where Queries

```
def query = Person.where {
     lastName == "Simpson"
}
def bartQuery = query.where {
     firstName == "Bart"
}
Person p = bartQuery.find()
```


Named Queries

- Allow you to define criteria queries as part of the domain class
- Can be chained together
- Can be combined with dynamic finders
- Break down queries into components

```
class Customer {
    //...
    static namedQueries = {
        goldLevelMoreThanFiveIncidents {
            byServiceLevelName('Gold')
            moreThanFiveIncidents()
        }

        moreThanFiveIncidents {
            sizeGt('incidents', 5)
        }

        byServiceLevelName { serviceLevelName ->
            serviceLevel {
                eq('name', serviceLevelName)
            }
        }
    }
```

Auto Timestamping

- Two special properties for domain classes
  - dateCreated
  - lastUpdated
- Must be nullable
- Will be set automatically when object is persisted to the database

---

Events

- beforeInsert - Executed before an object is initially persisted to the database
- beforeUpdate - Executed before an object is updated
- beforeDelete - Executed before an object is deleted
- beforeValidate - Executed before an object is validated
- afterInsert - Executed after an object is persisted to the database
- afterUpdate - Executed after an object has been updated
- afterDelete - Executed after an object has been deleted
- onLoad - Executed when an object is loaded from the database